Documentation on ALARO-0 developments in radiative scheme, based on cy32t1

Working material for ALARO Training Course,
Radostovice, 26.–30.3.2007
(compiled by Ján Mašek, last revised 13.4.2007)

1 Brief overview of the new developments

For the time being ALARO-0 developments in radiative scheme cover 3 areas:

1. Improved parameterization of cloud optical properties.

   Old ACRANEB scheme did not assume dependency of cloud absorption coefficient $k_{l|i}^{\mathrm{abs}}$, scattering coefficient $k_{l|i}^{\mathrm{scat}}$ and asymmetry factor $g_{l|i}$ on liquid/ice water content $\rho_{l|i}$. Moreover, only mean saturation effect was taken into account for clouds. With introduction of prognostic cloud water and ice it became highly desirable to relax these two simplifications. In ALARO-0 dependency of cloud optical properties on cloud water content $\rho_{l|i}$ as well as non-local saturation effect depending on cloud thickness and geometry are included. Parameterization scheme was developed and tuned in idealized framework, using experimental spectral data for 7 water and 16 ice clouds. Cloud optical saturation is assumed to be independent from gaseous one.

2. Introduction of Voigt effect in computation of gaseous transmissions.

   ACRANEB scheme assumes only 3 radiatively most important atmospheric gases (CO_2, H_2O and O_3). In original formulation equivalent band widths for these gases were computed using Malkmus band model. This model assumes Lorentz shape of absorption lines, caused by collision broadening. More realistic line shape is described by Voigt profile, which takes into account also Doppler broadening. Generalization of Malkmus formula in order to accommodate Voigt effect is therefore important for improving gaseous transmissions in upper atmosphere, where Doppler broadening becomes dominant process due to less frequent molecular collisions. But since it leads to rather expensive expression for equivalent band width as a function of absorber amount, there are two implementations in ALARO-0 – exact and approximate. In latter case the most expensive term is replaced by estimate of its global average.

3. New statistical model for bracketting technique in the NER (Net Exchanged Rate) formalism.

   Bracketting technique is used for better estimation of saturated layer optical thicknesses in thermal band, leading to more realistic energy exchanges due to absorption–emission. It is based on the fact that in gas-only atmosphere apparent optical thickness of given layer decreases with the distance from which the layer is viewed. Maximum optical thickness is obtained when the layer is viewed from its interface, minimum one when it is viewed either from surface or from the top of atmosphere. Actual value for any exchange lies between these two extremes and can in principle be obtained by interpolation with weight $0 \leq \alpha \leq 1$ (where 0 corresponds to minimum and 1 corresponds to maximum optical thickness).

In NER formalism primary exchanges (i.e. cooling to space, exchange with surface and exchange with adjacent layers) are computed as accurately as possible, and the treatment of remaining exchanges is simplified. Accurate computation of all exchanges is extremely costly (it requires evaluation of optical thicknesses for each pair of layers), but it enables to get reference solution which can be used offline for finding optimal interpolation weights α for secondary exchanges. It was seen that α increases in lower atmosphere as well as in the regions with temperature inversion.

In the first version, interpolation weight α was constructed as 2-parametric fit depending on height (via σ-coordinate p/p_S) and atmospheric stability (given by vertical gradient of potential temperature). Drawback of this simple fit was no vertical dependency of sensitivity to atmospheric stability, which led to slight problem in stratosphere. Problem was eliminated by introducing more sophisticated 6-parametric fit of α, which enables different treatment of tropospheric and stratospheric inversions. It was also recognized that α fits must be retuned more towards 'distant' (i.e. minimal) exchanges when the exact treatment of exchange between adjacent layers is turned on. This is especially true at the edges of atmosphere.

2 Basics of ACRANEB radiative transfer scheme

ACRANEB is an economical radiative transfer scheme which takes into account effect of relevant atmospheric gases, aerosols and clouds, with elaborated treatment of radiative saturation and multiple scattering. Due to efficiency reasons it splits electromagnetic spectrum only in two bands – solar and thermal. Even if radiative transfer in both bands is driven by the same fundamental principles, each band has its specificities. In solar band it is presence of direct flux (unscattered parallel radiation from the Sun) and absence of atmospheric emission. On the contrary, in thermal band direct flux is absent and atmospheric and surface emissions play an important role.

Radiative transfer equation used in ACRANEB is based on several simplifying assumptions:

- Each column is treated as plane parallel atmosphere split to J homogeneous layers. There is no lateral exchange with neighbouring columns.
- Diffuse fluxes are hemispheric constant, i.e. intensity (or radiance) I has the form:

$$I(\mu) = \begin{cases} I^+; & \mu < 0 \\ I^-; & \mu > 0 \end{cases} \qquad \mu = \cos\theta$$



Quantity μ denotes cosine of zenithal angle θ, it is measured positively downwards ($\mu = 1$ points to nadir, $\mu = -1$ points to zenith). Assumption of horizontal homogenity implies no dependency on azimuthal angle.

- Dependency of phase function P on scattering angle Θ is linear in $\cos\Theta$, with additional Dirac δ-treatment of forward scattering peak:

$$P(\cos\Theta) = 2f\delta(\cos\Theta - 1) + (1-f)\left(1 + 3\,\frac{g-f}{1-f}\cos\Theta\right)$$

Quantity g is asymmetry factor and f is relative strength of forward δ-peak. In ACRANEB scheme it is chosen as $f = g^2$. This choice assures that second moment of phase function P equals to that of Henyey-Greenstein phase function with the same asymmetry factor g (Henyey-Greenstein phase function is often used as 1-parametric approximation for Lorentz-Mie scattering).

For gases alone δ-scaling is not active, since Rayleigh scattering in forward and backward directions is symmetric ($g = 0$). This implies $f = 0$ and $P = 1$, i.e. Rayleigh scattering is approximated by isotropic phase function.

- Each layer is divided to clear sky and cloudy parts which are extending vertically through the whole layer. Lateral exchange between these two parts does not exist.
- Cloud overlaps between adjacent layers can be either random (`LRNUMX=.FALSE.`) or maximum (`LRNUMX=.TRUE.`).

When mentioned approximations are applied on fundamental equation of radiative transfer, they lead to following δ-two stream system:

$$\begin{aligned}
\frac{\mathrm{d}F^{\uparrow}}{\mathrm{d}\delta} &= 2[1-\varpi(1-\bar{\beta})]F^{\uparrow} - 2\varpi\bar{\beta}F^{\downarrow} - 2\pi(1-\varpi)B(T) - \varpi\beta(\mu_0)\frac{S}{\mu_0} \\
\frac{\mathrm{d}F^{\downarrow}}{\mathrm{d}\delta} &= 2\varpi\bar{\beta}F^{\uparrow} - 2[1-\varpi(1-\bar{\beta})]F^{\downarrow} + 2\pi(1-\varpi)B(T) + \varpi[1-\beta(\mu_0)]\frac{S}{\mu_0} \\
\frac{\mathrm{d}S}{\mathrm{d}\delta} &= -\frac{S}{\mu_0}
\end{aligned}$$

Here S denotes direct flux (unscattered parallel radiation coming from the Sun), $F^{\uparrow}$ and $F^{\downarrow}$ are upward and downward diffuse fluxes, $B(T)$ is intensity of blackbody radiation, δ is optical depth measured from top of atmosphere downwards, ϖ is single scattering albedo, $\bar{\beta}$ is backscatter fraction for diffuse fluxes and $\beta(\mu_0)$ is upscatter fraction for direct flux with μ_0 being cosine of solar zenithal angle:

$$\begin{aligned}
\mathrm{d}\delta &= \rho_r(k^{\mathrm{abs}} + k^{\mathrm{scat}})\,\mathrm{d}z \qquad & \beta(\mu_0) &= \frac{1}{2} - \frac{3}{4}\cdot\frac{g}{1+g}\,\mu_0 \\
\varpi &= \frac{k^{\mathrm{scat}}}{k^{\mathrm{abs}} + k^{\mathrm{scat}}} & \bar{\beta} &= \int_0^1 \beta(\mu)\,\mathrm{d}\mu = \frac{4+g}{8(1+g)}
\end{aligned}$$

Density ρ_r refers to radiatively active matter (gases, aerosols or clouds). Coefficients k^{abs}, k^{scat} have units $\mathrm{m^2\,kg^{-1}}$. In model, however, they are divided by gravity acceleration so that units change to $\mathrm{Pa^{-1}}$.

Because of Dirac δ-treament of forward scattering peak, coefficients k^{abs}, k^{scat} and derived quantities occuring in δ-two stream system are rescaled correspondingly:

$$\begin{aligned}
k^{\mathrm{abs}} &\mapsto k^{\mathrm{abs}} & \mathrm{d}\delta &\mapsto (1-\varpi f)\,\mathrm{d}\delta \\
k^{\mathrm{scat}} &\mapsto (1-f)k^{\mathrm{scat}} & \varpi &\mapsto \frac{1-f}{1-\varpi f}\,\varpi
\end{aligned}$$

This transformation is based on scaling invariance of radiative transfer equation. In order to remain valid also in presence of direct flux, definition of this flux must be slightly modified: direct flux scattered in forward direction ($\Theta = 0$) is assumed to remain direct, instead of being converted to diffuse flux. Even with this modification scaling invariance holds only approximately (it is exact for $\mu_0 = \frac{1}{2}$).

Obtained δ-two stream system can be written in more compact form. For example, in solar band one neglects emission term containing $B(T)$ and defines auxiliary quantities $\alpha_1, \ldots, \alpha_5$:

$$\alpha_1 = 2[1 - \varpi(1 - \bar{\beta})] \qquad \alpha_3 = \varpi\beta(\mu_0) \qquad \alpha_5 = \frac{1}{\mu_0}$$
$$\alpha_2 = 2\varpi\bar{\beta} \qquad \alpha_4 = \varpi[1 - \beta(\mu_0)]$$

With these notations system can be rewritten into the form:

$$\frac{\mathrm{d}}{\mathrm{d}\delta}\begin{bmatrix} F^{\uparrow} \\ F^{\downarrow} \\ S \end{bmatrix} = \begin{bmatrix} \alpha_1 & -\alpha_2 & -\alpha_3\alpha_5 \\ \alpha_2 & -\alpha_1 & \alpha_4\alpha_5 \\ 0 & 0 & -\alpha_5 \end{bmatrix} \cdot \begin{bmatrix} F^{\uparrow} \\ F^{\downarrow} \\ S \end{bmatrix}$$

For homogeneous layer j coefficients $\alpha_{1j}, \ldots, \alpha_{5j}$ do not depend on optical depth δ and the system can be integrated analytically, giving linear relation between incoming and outgoing fluxes:

$$\begin{bmatrix} S_B \\ F_B^{\downarrow} \\ F_T^{\uparrow} \end{bmatrix}_j = \begin{bmatrix} a_1 & 0 & 0 \\ a_2 & a_4 & a_5 \\ a_3 & a_5 & a_4 \end{bmatrix}_j \cdot \begin{bmatrix} S_T \\ F_T^{\downarrow} \\ F_B^{\uparrow} \end{bmatrix}_j$$

Subscripts B and T stand for layer bottom and top. Analytical expressions for transmissivities a_1, a_2, a_4 and reflectivities a_3, a_5 can be found e.g. in [1].

Slicing atmosphere into J homogeneous layers and equating fluxes outgoing from given layer to fluxes entering neighbouring layers is referred to as adding method. It leads to the linear system of $3J$ equations for $3J+3$ unknowns (in solar band). Matrix of the system has band structure with 5 non-zero diagonals. In order to close the system, 3 additional equations are needed. These come from boundary conditions:

1. prescribed direct flux S at the top of atmosphere
2. zero downward diffuse flux $F^{\downarrow}$ at the top of atmosphere
3. reflection condition at the Earth's surface (with $A(\mu_0)$ and $\bar{A}$ being surface albedos for direct and diffuse fluxes):

$$F^{\uparrow}_{\mathrm{surf}} = A(\mu_0)S_{\mathrm{surf}} + \bar{A}F^{\downarrow}_{\mathrm{surf}}$$

Thanks to band matrix system can be solved very efficiently, with cost being proportional to J. In the presence of cloudiness procedure becomes more complicated, but the basic principle of adding remains. One must assume two sets of fluxes – first in clear sky part, second in cloudy part of the layers. At layer interfaces fluxes from both parts must be redistributed according to chosen cloud overlapping assumption.

All what was said so far is valid only for monochromatic computations or in grey body case (optical properties independent on wavelength). Main problem in ACRANEB scheme comes from the use of wide spectral intervals. Gaseous absorption lines prevent the use of grey body approximation even for relatively narrow spectral bands. In broadband approach, grey body approximation loses its validity also for clouds. Problem is that dependency of transmissivities and reflectivities on coefficients k^{abs}, k^{scat} and g is highly nonlinear, so that averaged values $\bar{a}_1, \ldots, \bar{a}_5$ can differ considerably from values diagnosed from averages $\overline{k^{\mathrm{abs}}}$, $\overline{k^{\mathrm{scat}}}$ and $\bar{g}$ (this is especially true for large optical depths). Moreover, averaging procedure depends on spectral composition of fluxes, which in turn depends on photon paths influenced by multiple scattering. For these reasons broadband approach requires parametrization of radiative saturation for gases and clouds. Such parameterization usually consists of two parts:

1. Fitting of saturated broadband coefficients k^{abs}, k^{scat} and g to values obtained by very precise and costly wavelength by wavelength computations. For gases these computations are usually done in idealized non-scattering case.

2. More ad hoc treatment which should roughly account for the effect of multiple scattering.

3 Localization of the code

All concerned pieces of code are related through the following calling tree:

```
APLPAR
  |
  ACRANEB (clear sky fluxes)
  | |
  | AC_CLOUD_MODEL
  |
  ACNEBN
  |
  ACRANEB (full computation including clouds)
    |
    AC_CLOUD_MODEL
```

Radiative fluxes are computed in `ACRANEB`, which is called from two places in `APLPAR`. First call is invoked only when clear sky fluxes are asked. Second call is placed after cloud fraction `PNEB` and specific mass of cloud water `PQLI` and cloud ice `PQICE` were diagnosed in `ACNEBN`. `ACRANEB` calls new subroutine `AC_CLOUD_MODEL`, which determines cloud optical properties. Call of this subroutine in clear sky case is useless.

Before the first call to `ACRANEB`, vertical profiles `ZMAN` and `ZMAK` are precomputed in `APLPAR`. They are passed to `ACRANEB` where they are used to compute statistical weights `ZMIXP`, needed for bracketting technique in NER formalism.

4 Description of changes

4.1 Improved cloud optical properties

Concerned subroutines:

```
APLPAR
ACRANEB
AC_CLOUD_MODEL
```

Driving logical keys:

`LCLSATUR` – activates cloud saturation and dependency of $k^{\mathrm{abs}}_{l|i}$, $k^{\mathrm{scat}}_{l|i}$ and $g_{l|i}$ on $\rho_{l|i}$

Changes in `APLPAR`:

Array `PR` (gas constant of air R) is passed to `ACRANEB`.

Changes in ACRANEB:

At the very beginning of ACRANEB call to AC_CLOUD_MODEL was inserted. New input argument PR had to be introduced, since it is needed in AC_CLOUD_MODEL to diagnose air density ρ and to convert specific mass of cloud water and ice $q_{l|i}$ to corresponding density $\rho_{l|i}$. Subroutine AC_CLOUD_MODEL returns 2D arrays with optical properties of ice and liquid clouds, dimensioned as (KLON,KLEV):

ZBSF{S,T}{I,N}	– back scatter fraction $\bar{\beta}$	[1]
ZEOA{S,T}{I,N}	– absorption coefficient k^{abs}	[Pa^{-1}]
ZEOD{S,T}{I,N}	– scattering coefficient k^{scat}	[Pa^{-1}]
ZUSA{I,N}	– coefficient a for computing upscatter fraction $\beta(\mu_0)$	[1]
ZUSB{I,N}	– coefficient b for computing upscatter fraction $\beta(\mu_0)$	[1]

S,T stands for solar and thermal band, while I,N denotes ice and liquid clouds (letter N comes from French 'nuage').

In old version of ACRANEB variables ZEO1T{I,N}, ZEO2T{I,N} (coefficients α_1, α_2) did not depend on location and were evaluated directly from scalar namelist values BSFT{I,N}, EOAT{I,N} and EODT{I,N}. This is no more true, now they must be precomputed as 2D arrays:

```
DO JLEV=KTDIA,KLEV
  DO JLON=KIDIA,KFDIA
    ZEO2TI(JLON,JLEV)=2._JPRB*ZBSFTI(JLON,JLEV)*ZEODTI(JLON,JLEV)
    ZEO2TN(JLON,JLEV)=2._JPRB*ZBSFTN(JLON,JLEV)*ZEODTN(JLON,JLEV)
    ZEO1TI(JLON,JLEV)=ZEO2TI(JLON,JLEV)+2._JPRB*ZEOATI(JLON,JLEV)
    ZEO1TN(JLON,JLEV)=ZEO2TN(JLON,JLEV)+2._JPRB*ZEOATN(JLON,JLEV)
  ENDDO
ENDDO
```

Later in thermal computations these arrays are used to update values ZEO1, ZEO2 in order to get final layer quantities $\alpha_1\Delta\delta$, $\alpha_2\Delta\delta$ for gases + aerosols + liquid clouds + ice clouds:

```
ZEO1=ZEO1+ZEO1TN(JLON,JLEV)*(PDELP(JLON,JLEV)*ZQLI(JLON,JLEV))&
 &+ZEO1TI(JLON,JLEV)*(PDELP(JLON,JLEV)*ZQICE(JLON,JLEV))
ZEO2=ZEO2+ZEO2TN(JLON,JLEV)*(PDELP(JLON,JLEV)*ZQLI(JLON,JLEV))&
 &+ZEO2TI(JLON,JLEV)*(PDELP(JLON,JLEV)*ZQICE(JLON,JLEV))
```

In solar computations upscatter fractions ZUS{I,N} are no longer only functions of geographical location, but via coefficients ZUSA{I,N} they depend also on vertical level (coefficients ZUSB{I,N} are set to zero in AC_CLOUD_MODEL, but they were kept for generality). This is why they are diagnosed directly inside main JLEV, JLON loop, together with quantities ZEO1S{I,N}, ZEO2S{I,N}:

```
ZEO2SN=2._JPRB*ZBSFSN(JLON,JLEV)*ZEODSN(JLON,JLEV)
ZEO2SI=2._JPRB*ZBSFSI(JLON,JLEV)*ZEODSI(JLON,JLEV)
ZEO1SN=ZEO2SN+2._JPRB*ZEOASN(JLON,JLEV)
ZEO1SI=ZEO2SI+2._JPRB*ZEOASI(JLON,JLEV)
ZEO1=ZEO1+ZEO1SN*(PDELP(JLON,JLEV)*ZQLI(JLON,JLEV))&
 &+ZEO1SI*(PDELP(JLON,JLEV)*ZQICE(JLON,JLEV))
ZEO2=ZEO2+ZEO2SN*(PDELP(JLON,JLEV)*ZQLI(JLON,JLEV))&
 &+ZEO2SI*(PDELP(JLON,JLEV)*ZQICE(JLON,JLEV))
...
```

```
ZEOSN=ZEODSN(JLON,JLEV)+ZEOASN(JLON,JLEV)
ZEOSI=ZEODSI(JLON,JLEV)+ZEOASI(JLON,JLEV)
ZEO=ZEO+ZEOSN*(PDELP(JLON,JLEV)*ZQLI(JLON,JLEV))&
 &+ZEOSI*(PDELP(JLON,JLEV)*ZQICE(JLON,JLEV))
...
ZUSN=(0.5_JPRB+ZUSAN(JLON,JLEV)*ZMUO(JLON))/&
 &(1._JPRB+ZUSBN(JLON,JLEV)*ZMUO(JLON))
ZUSI=(0.5_JPRB+ZUSAI(JLON,JLEV)*ZMUO(JLON))/&
 &(1._JPRB+ZUSBI(JLON,JLEV)*ZMUO(JLON))
ZEO3=(ZEO3/ZMUOIC+ZUSN*((PDELP(JLON,JLEV)&
 &*ZQLI(JLON,JLEV))*ZEODSN(JLON,JLEV))&
 &+ZUSI*((PDELP(JLON,JLEV)&
 &*ZQICE(JLON,JLEV))*ZEODSI(JLON,JLEV)))*ZMUOIN
ZEO4=(ZEO4/ZMUOIC+(1._JPRB-ZUSN)*((PDELP(JLON,JLEV)&
 &*ZQLI(JLON,JLEV))*ZEODSN(JLON,JLEV))&
 &+(1._JPRB-ZUSI)*((PDELP(JLON,JLEV)&
 &*ZQICE(JLON,JLEV))*ZEODSI(JLON,JLEV)))*ZMUOIN
ZEO5=ZEO*ZMUOIN
```

In given piece of code it can also be seen how variables `ZEO1`, ..., `ZEO5` and `ZEO` are updated in order to get final layer quantities $\alpha_1\Delta\delta$, ..., $\alpha_5\Delta\delta$ and $\Delta\delta$, valid again for gases + aerosols + liquid clouds + ice clouds.

Namelist variables `EOA{S,T}{I,N}`, `EOD{S,T}{I,N}`, `BSF{S,T}{I,N}`, `USA{I,N}` and `USB{I,N}` from module `YOMPHY3` were removed from `ACRANEB`. They are still needed in `AC_CLOUD_MODEL`.

Design of `AC_CLOUD_MODEL`:

`AC_CLOUD_MODEL` is a new subroutine used for computation of cloud optical properties. For the case `LCLSATUR=.FALSE.` it reproduces old computation by setting output arrays to constant namelist values:

```
DO JLEV = KTDIA, KLEV
  DO JLON = KIDIA, KFDIA

    ! absorption coefficient
    PEOASI(JLON, JLEV) = EOASI
    ...
    ! scattering coefficient
    PEODSI(JLON, JLEV) = EODSI
    ...
    ! back scatter fraction
    PBSFSI(JLON, JLEV) = BSFSI
    ...
    ! coefficients for computation of upscatter fraction
    PUSAI (JLON, JLEV) = USAI
    ...
  ENDDO
ENDDO
```

Rest of the subroutine deals with the new cloud treatment (`LCLSATUR=.TRUE.`). It is designed in a modular way which enables easy generalization to more than 2 spectral

bands (all spectral computations are inside `JB` loop running through spectral bands). The only restriction is given by output arrays explicitly divided to solar and thermal quantities. This was necessary for compatibility with current `ACRANEB` code.

Quantities common for both spectral bands are precomputed outside of `JB` loop. These are geometry factors needed for computation of effective optical depth, and scaled liquid/ice water content used for Pade fits of coefficients $k_{l|i}^{\mathrm{abs}}$, $k_{l|i}^{\mathrm{scat}}$ and $g_{l|i}$. Geometry factors $f(n_j, n_k)$ are stored in array `ZGEOM` and they depend on cloud fraction n at levels j and k:

$$f(n_j, n_k) = \begin{cases} (n_k)^p & \text{; random overlaps} \\ \min\left[1, \left(\dfrac{n_k}{n_j}\right)^p\right] & \text{; maximum overlaps} \end{cases} \qquad j \neq k$$

$$f(n_j, n_k) = 1 \qquad j = k$$

Exponent p is empirical tuning factor given via namelist variable `REXP_NEB`. For $p = 1$ factor $f(n_j, n_k)$ can be interpreted as fraction of cloud layer j overlapped by cloud layer k. Idealized tests showed that optimal value of p is around 8, which is default.

Scaled liquid/ice water content `Z{L,I}WC1` is diagnosed from air density `ZRHO` and specific mass of liquid/ice inside the cloud `PQ{L,I}`. In order to save some exponentiations, computation is done only if given layer contains cloud water/ice, which is indicated by logical array `LLQ{L,I}` (some of subsequent code examples will be restricted to liquid phase `L`):

```
IF ( LLQL(JLON, JLEV) ) THEN
  ZLWC1(JLON, JLEV) = (ZRHO*PQL(JLON, JLEV))**FCM_N_L
ELSE
  ZLWC1(JLON, JLEV) = 0._JPRB
ENDIF
```

Scaling exponents `FCM_N_{L,I}` had to be introduced in order to ensure sufficient accuracy of subsequent Pade fits. Without suitable scaling, order of the fits for some quantities would have to be higher than $(3, 3)$, which would be difficult to manage since there must not be positive roots in denominator.

Loop through spectral bands follows. First it fits scaled coefficients $k_{l|i}^{\mathrm{abs}}$, $k_{l|i}^{\mathrm{scat}}$ and $g_{l|i}$ to scaled liquid/ice water content $\rho_{l|i}$ using Pade approximants of the form:

$$y(x) = \frac{p_0 + p_1 x + p_2 x^2 + p_3 x^3}{1 + q_1 x + q_2 x^2 + q_3 x^3}$$

Fitting is done in private subroutine `FIT1`, which uses Pade coefficients p_i and q_i stored in arrays `FCM_P_{A,D,G}{L,I}` and `FCM_Q_{A,D,G}{L,I}` (`A,D,G` stands for absorption coefficient k^{abs}, scattering/diffusion coefficient k^{scat} and asymmetry factor g). They are dimensioned `(N_SPBAND,0:3)` and `(N_SPBAND,1:3)`, where first index denotes spectral band (1 – solar, 2 – thermal) and second index denotes order of Pade term.

Dependency of coefficients $k_{l|i}^{\mathrm{abs}}$, $k_{l|i}^{\mathrm{scat}}$ and $g_{l|i}$ on liquid/ice water content $\rho_{l|i}$ was derived from experimentally measured cloud data. However, direct use of asymmetry factor g obtained in this way lead to overestimation of cloud albedo in real case experiments. Possible reason might be complex geometry and inhomogenity of real clouds, which probably leads to higher transmissivity and smaller reflectivity when compared to idealized plane parallel case. Therefore, it was decided to enhance forward

scattering by reducing $1-g$ to 80 % of its original value. Pade fits provide already modified value of asymmetry factor g.

After fitting, arrays `ZEO{A,D}{L,I}` are unscaled using exp function and divided by gravity acceleration, in order to convert units of $k_{l|i}^{\mathrm{abs}}$, $k_{l|i}^{\mathrm{scat}}$ from $\mathrm{m^2\,kg^{-1}}$ to $\mathrm{Pa^{-1}}$. At the same time, asymmetry factor `ZG{L,I}` is unscaled using tanh type function which guarantees result from interval $(0, 1)$:

```
IF ( LLQL(JLON, JLEV) ) THEN
  ZEOAL(JLON, JLEV) = EXP(ZEOAL(JLON, JLEV))*ZRRG
  ZEODL(JLON, JLEV) = EXP(ZEODL(JLON, JLEV))*ZRRG
  ZGL  (JLON, JLEV) = &
    & 1.0_JPRB/( 1.0_JPRB + EXP(-2.0_JPRB*ZGL(JLON, JLEV)) )
ELSE
  ZEOAL(JLON, JLEV) = ZDRYEOAL
  ZEODL(JLON, JLEV) = ZDRYEODL
  ZGL  (JLON, JLEV) = ZDRYGL
ENDIF
```

It can be seen that in case of layer without cloud liquid/ice, coefficients are set to their dry asymptotic values `ZDRYEO{A,D}{L,I}` and `ZDRYG{L,I}`, precomputed at the beginning of `JB` loop:

```
ZDRYEOAI = EXP(FCM_P_AI(JB, 0))*ZRRG
ZDRYEOAL = EXP(FCM_P_AL(JB, 0))*ZRRG
ZDRYEODI = EXP(FCM_P_DI(JB, 0))*ZRRG
ZDRYEODL = EXP(FCM_P_DL(JB, 0))*ZRRG
ZDRYGI = &
  & 1.0_JPRB/( 1.0_JPRB + EXP(-2.0_JPRB*FCM_P_GI(JB, 0)) )
ZDRYGL = &
  & 1.0_JPRB/( 1.0_JPRB + EXP(-2.0_JPRB*FCM_P_GL(JB, 0)) )
```

Here the property $y(0) = p_0$ of Pade approximants is used. Precomputation of dry values outside of `JLEV`, `JLON` loop can save some exponentials.

Unscaled coefficients k^{abs}, k^{scat} enable to evaluate unsaturated optical depth δ_0 of each cloud layer (array `ZDEL0`):

```
ZDEL0(JLON, JLEV) = PDELP(JLON, JLEV)*( &
 & PQI(JLON, JLEV)*(ZEOAI(JLON, JLEV) + ZEODI(JLON, JLEV)) + &
 & PQL(JLON, JLEV)*(ZEOAL(JLON, JLEV) + ZEODL(JLON, JLEV)) )
```

As the next step, effective optical depth δ_0^{eff} for each cloud layer is computed, taking into account influence of other cloud layers:

$$\delta_{0j}^{\mathrm{eff}} = \sum_k f(n_j, n_k)\delta_{0k} = \delta_{0j} + \sum_{k\neq j} f(n_j, n_k)\delta_{0k}$$

Vertical summation goes either from the top to current cloud layer (array `ZDEL0_EFF`), or through all atmosphere (array `ZDEL0_EFF2`).

Cloud saturation is parameterized via saturation factors c^{abs} and c^{scat}, fitted to effective optical depth δ_0^{eff}. Dependency has the form:

$$c(\delta_{0j}^{\mathrm{eff}}) = \frac{1}{1 + \left(\dfrac{\delta_{0j}^{\mathrm{eff}}}{\delta_0^{\mathrm{crit}}}\right)^{\mu}}$$

Two fitting parameters are critical optical depth δ_0^{crit} for which $c = \frac{1}{2}$ (array `FCM_DEL_{A,D}`), and exponent μ determining sharpness of saturation profile (array `FCM_MU_{A,D}`). Their optimal values were found in idealized environment, performing wavelength by wavelength computations for sample of 1-layer homogeneous clouds with different geometrical thicknesses and liquid/ice water contents. Requirement that saturated optical depth of homogeneous cloud $\delta = c(\delta_0)\delta_0$ increases with unsaturated optical depth δ_0 imposes restriction $0 < \mu \leq 1$.

For solar absorption, above mentioned dependency is modified in order to better describe vertical profile of saturation inside cloud. Simplified considerations dealing with pure absorption and pure scattering in homogeneous rectangular cloud showed that while saturation factor c^{scat} does not depend on height (which justifies summation of δ_0^{eff} through all atmosphere), there is significant vertical variation of saturation factor c^{abs} (this is due to the fact that in pure absorption case solar flux at given level is influenced by cloud layers above, but not by cloud layers below – provided that surface albedo is not too high). It can be shown easily that in the absence of scattering and surface reflection, vertical variation of saturation factor c^{abs} can be obtained from previous formula as:

$$\tilde{c}(\delta_0) = \frac{\mathrm{d}}{\mathrm{d}\delta_0}[c(\delta_0)\delta_0] = c(\delta_0)[\mu c(\delta_0) + 1 - \mu]$$

In the new formula, δ_0 denotes effective optical depth summed from the top to current cloud layer.

For efficiency reasons, power function occuring in saturation factors `ZCA`, `ZCD` is decomposed to exponential and logarithm:

```
! saturation factors c_abs, c_scat
ZARG2 = LOG(MAX(ZDEL0_EFF2(JLON, JLEV), ZEPS))
IF ( JB == 1 ) THEN

  ! solar band
  ZARG = LOG(MAX(ZDEL0_EFF(JLON, JLEV), ZEPS))
  ZCA = 1.0_JPRB/(1.0_JPRB + EXP(ZMUA*ZARG  - ZDELA))
  ZCA = ZCA*(ZMUA*ZCA + 1.0_JPRB - ZMUA)
  ZCD = 1.0_JPRB/(1.0_JPRB + EXP(ZMUD*ZARG2 - ZDELD))

ELSE

  ! thermal band
  ZCA = 1.0_JPRB/(1.0_JPRB + EXP(ZMUA*ZARG2 - ZDELA))
  ZCD = 1.0_JPRB/(1.0_JPRB + EXP(ZMUD*ZARG2 - ZDELD))

ENDIF
```

Auxiliary quantities `ZMU{A,D}` and `ZDEL{A,D}` are precomputed at the beginning of `JB` loop:

```
ZMUA  = FCM_MU_A(JB)
ZMUD  = FCM_MU_D(JB)
ZDELA = ZMUA*LOG(FCM_DEL_A(JB))
ZDELD = ZMUD*LOG(FCM_DEL_D(JB))
```

Local saturation factors `ZCA` and `ZCD` are applied to absorption and scattering coefficients in order to get their saturated values:

```
! saturated absorption coefficient k_abs
ZEOAI(JLON, JLEV) = ZCA*ZEOAI(JLON, JLEV)
ZEOAL(JLON, JLEV) = ZCA*ZEOAL(JLON, JLEV)

! saturated scattering coefficient k_scat
ZEODI(JLON, JLEV) = ZCD*ZEODI(JLON, JLEV)
ZEODL(JLON, JLEV) = ZCD*ZEODL(JLON, JLEV)
```

At the end of JB loop, output arrays PEO{A,D}{S,T}{L,I} and PBSF{S,T}{L,I} are filled. Back scatter fraction is diagnosed from asymmetry factor ZG{L,I} using formula $\bar{\beta} = (4+g)/(8+8g)$:

```
IF ( JB == 1 ) THEN

  ! solar band
  DO JLEV = KTDIA, KLEV
    DO JLON = KIDIA, KFDIA
      PEOASI(JLON, JLEV) = ZEOAI(JLON, JLEV)
      PEOASL(JLON, JLEV) = ZEOAL(JLON, JLEV)
      PEODSI(JLON, JLEV) = ZEODI(JLON, JLEV)
      PEODSL(JLON, JLEV) = ZEODL(JLON, JLEV)
      PBSFSI(JLON, JLEV) = (4.0_JPRB + ZGI(JLON, JLEV))/ &
       & (8.0_JPRB + 8.0_JPRB*ZGI(JLON, JLEV))
      PBSFSL(JLON, JLEV) = (4.0_JPRB + ZGL(JLON, JLEV))/ &
       & (8.0_JPRB + 8.0_JPRB*ZGL(JLON, JLEV))

      ! coefficients for computation of upscatter fraction
      PUSAI(JLON, JLEV) = 2.0_JPRB*PBSFSI(JLON, JLEV) - 1.0_JPRB
      PUSAL(JLON, JLEV) = 2.0_JPRB*PBSFSL(JLON, JLEV) - 1.0_JPRB
      PUSBI(JLON, JLEV) = 0.0_JPRB
      PUSBL(JLON, JLEV) = 0.0_JPRB
    ENDDO
  ENDDO

ELSEIF ( JB == 2 ) THEN

  ! thermal band
  DO JLEV = KTDIA, KLEV
    DO JLON = KIDIA, KFDIA
      PEOATI(JLON, JLEV) = ZEOAI(JLON, JLEV)
      PEOATL(JLON, JLEV) = ZEOAL(JLON, JLEV)
      PEODTI(JLON, JLEV) = ZEODI(JLON, JLEV)
      PEODTL(JLON, JLEV) = ZEODL(JLON, JLEV)
      PBSFTI(JLON, JLEV) = (4.0_JPRB + ZGI(JLON, JLEV))/ &
       & (8.0_JPRB + 8.0_JPRB*ZGI(JLON, JLEV))
      PBSFTL(JLON, JLEV) = (4.0_JPRB + ZGL(JLON, JLEV))/ &
       & (8.0_JPRB + 8.0_JPRB*ZGL(JLON, JLEV))
    ENDDO
  ENDDO

ENDIF
```

It can be seen that in solar band aditional arrays PUSA{L,I} and PUSB{L,I} are computed. These are coefficients a and b which are used in general formula for upscatter fraction $\beta(\mu_0) = (\frac{1}{2} + a\mu_0)/(1 + b\mu_0)$. Their setting comes from special relation between upscatter fraction $\beta(\mu_0)$ and backscatter fraction $\bar{\beta}$, which holds in used δ-two stream system:

$$\beta(\mu_0) = \frac{1}{2} + (2\bar{\beta} - 1)\mu_0$$

Finally it should be noted that default values of namelist variables FCM_* (FCM stands for 'Fits for Cloud Model') were tuned in idealized cloud simulation model based on experimentally measured cloud spectral properties. Most of them should not be changed by hand, since without complex retuning one is likely to introduce inconsistencies. What can be played with are parameters FCM_DEL_{A,D}, FCM_MU_{A,D} appearing in fits of saturation factors c^{abs}, c^{scat}. Effect of cloud geometry can be fine tuned by changing scaling exponent for cloud fraction REXP_NEB.

4.2 Voigt effect

Concerned subroutines:

APLPAR
ACRANEB

Driving logical keys:

LVOIGT – activates treatment of Voigt effect
LVFULL – activates exact computation (expensive)

Changes in APLPAR:

Array PR (gas constant of air R) is passed to ACRANEB.

Changes in ACRANEB:

Treatment of Voigt effect modifies computation of equivalent band width ZW{S,T}{C,H,O} (S,T denotes solar and thermal band; C,H,O denotes CO_2, H_2O and O_3). It is introduced via correction factor ZAFVOI which is inserted into formula based on Malkmus band model (value 1 means unmodified Malkmus formula). Correction is inserted on every place where equivalent band width is computed. Following example illustrates it for H_2O in thermal band (ZWTH):

```
ZAFVOI=1._JPRB
IF (LVOIGT) THEN
  IF (LVFULL) THEN
    ZVOIEMP=EXP(ZEPSV*LOG(ZIBV0*(ZBZV/ZVOIGT))*LOG(ZIZV0*ZVOIGT))
  ELSE
    ZVOIEMP=ZVOISIM
  ENDIF
  ZAFVOI=1._JPRB+ZVOIGT/(RPI*(ZBZV/ZVOIGT)/(1.5_JPRB*SQRT(ZBZV))&
   &+4._JPRB/(ZBZV/ZVOIGT)+5._JPRB+ZGAMV*SQRT(ZBZV/ZVOIGT)&
   &*ZVOIEMP)
ENDIF
ZWTH=(ZGAS2B(4)*(ZRTH(JLON)/ZNTH(JLON)))*(SQRT(1._JPRB+ZAFVOI&
 &*ZBZV)-1._JPRB)+GCC(4)*ZCTH(JLON)
```

Exact computation (LVFULL=.TRUE.) evaluates term ZVOIEMP using exponential and two logarithms. Since this is done inside JLEV, JLON loop, computation is expensive and cheaper alternative had to be implemented (branch LVFULL=.FALSE.). It simply sets ZVOIEMP to constant value ZVOISIM, which is the estimate of its global average (currently set to 1.21).

Computation of value ZVOIGT requires air density. For this reason subroutine interface had to be extended by input array PR (gas constant of air R). Air density can then be diagnosed from state equation.

4.3 Statistical model for bracketting technique in the NER formalism

Concerned subroutines:

APLPAR
ACRANEB

Driving logical keys:

LRMIX – activates use of bracketting technique
LREWS – activates exact computation of exchange with surface
LRPROX – activates exact computation of exchange between adjacent layers
LRAUTOEV – activates exact computation of all exchanges (expensive)
LNEWSTAT – activates new statistical fit of weights α

Changes in APLPAR:

Auxiliary profiles ZMAN, ZMAK for determination of weights α are precomputed, then they are passed to ACRANEB. Since these profiles do not depend on geographical location, arrays ZMAN, ZMAK are dimensioned as 0:KLEV, but only elements 1:KLEV-1 are initialized. For old statistical fit (LNEWSTAT=.FALSE.) following expressions are used:

$$\begin{aligned} \texttt{ZMAN(JLEV)} &= 0.3\sigma_j^* \\ \texttt{ZMAK(JLEV)} &= 0.1 \end{aligned}$$

Index j denotes half level JLEV, $\sigma_j^* = p_j^*/p_{\mathrm{ref}}$ where p_j^* is corresponding pressure in standard atmosphere and $p_{\mathrm{ref}} = 101\,325\,\mathrm{Pa}$.

New statistical fit (LNEWSTAT=.TRUE.) is more sophisticated and it uses different settings for approximate and exact treatment of exchange between adjacent layers. In case of approximate treatment (LRPROX=.FALSE.) it sets:

$$\begin{aligned} \texttt{ZMAN(JLEV)} &= 0.29\tanh(2.0\sigma_j^*) \\ \texttt{ZMAK(JLEV)} &= 0.07{\sigma_j^*}^{1.5} + 0.14(1-\sigma_j^*)^{20} \end{aligned}$$

For exact treatment (LRPROX=.TRUE.) tuning of ZMAN is different:

$$\begin{aligned} \texttt{ZMAN(JLEV)} &= 0.14\tanh(3.5\sigma_j^*) \\ \texttt{ZMAK(JLEV)} &= 0.07{\sigma_j^*}^{1.5} + 0.14(1-\sigma_j^*)^{20} \end{aligned}$$

Moreover, in this case values ZMAK close to the edges of atmosphere must be pushed toward zero in order to prevent double accounting of 'local' effect:

```
IF ( LRPROX ) THEN
  DO JLEV = KTDIA, KTDIA+3
    ZMAK(JLEV) = ZMAK(JLEV) / 2**(KTDIA+4-JLEV)
  ENDDO
  DO JLEV = KLEV-4, KLEV-1
    ZMAK(JLEV) = ZMAK(JLEV) / 2**(5-KLEV+JLEV)
  ENDDO
ENDIF
```

Changes in `ACRANEB`:

Subroutine interface was extended by input arrays `PMAK`, `PMAN`, precomputed in `APLPAR`. Computation of weights α_j (array `ZMIXP`) for half levels `1:KLEV-1` now reads:

$$\alpha_j = \min\left[1, \texttt{PMAN(JLEV)} + \texttt{PMAK(JLEV)} \cdot \max\left(0, \frac{\Delta(c_p\tilde{\theta})_j}{\Delta\phi_j}\right)\right]$$

$$\tilde{\theta} = T\left(\frac{p_S}{p}\right)^{\kappa} \qquad \kappa = \frac{R}{c_p}$$

Symbols R and c_p denote gas constant of air and specific heat of air at constant pressure, ϕ is geopotential. Weights α_j depend on geographical location via atmospheric stability $\frac{\partial(c_p\tilde{\theta})}{\partial\phi}$, where quantity $\tilde{\theta}$ in given profile is proportional to potential temperature (difference is caused by the use of surface pressure p_S in its definition, instead of constant reference pressure 100 000 Pa). At the model top ($j = 0$) and bottom ($j = J$) weights are extrapolated:

$$\alpha_0 = \alpha_1$$
$$\alpha_J = \alpha_{J-1}$$

References

[1] Geleyn, J.-F., and A. Hollingsworth, 1979: An economical analytical method for the computation of the interaction between scattering and line absorption of radiation. *Contrib. Atmos. Phys.*, **52**, 1–16.

[2] Ritter, B., and J.-F. Geleyn, 1992: A comprehensive radiation scheme for numerical weather prediction models with potential applications in climate simulations. *Mon. Wea. Rev.*, **120**, 303–325.

[3] Gérard, L., 2001: ALADIN operational physics documentation. Part VIII – Radiation and chemicals. *Available from ALADIN web page.*

[4] Geleyn, J.-F., R. Fournier, G. Hello, and N. Pristov, 2005: A new 'bracketing' technique for flexible and economical computation of thermal radiative fluxes, scattering effects included, on the basis of the Net Exchanged Rate (NER) formalism. *WGNE Blue Book, 2005 Edition.*

[5] Geleyn, J.-F., P. Bénard, and R. Fournier, 2005: A general-purpose extension of the Malkmus band-model average equivalent width to the case of the Voigt line profile. *Q. J. R. Meteorol. Soc.*, **131**, 2757–2768.